
Campos Vetoriais no plano; Integrais de Linha
Para funções vetoriais, vamos precisar:

> restart:

> with(linalg):

Warning, new definition for norm

Warning, new definition for trace

Plotando campos vetoriais no plano:

Para fazer isto, precisamos do comando "fieldplot" do biblioteca "plots" :

> with(plots,fieldplot);

> vf:=[-y,x];

Istp define o campo de vetores -y i + x j ; os colchetes sao exigidos e as componentes 
podem ser qualquer expressao em x and y (outras variaveis podem ser usadas, eh claro).

> fieldplot(vf,x=-2..2,y=-2..2);



Uma questão comum é perguntar é saber se um campo é gradiente de uma função. Se um 
campo de vetores não tem singularidades (pontos onde uma ou ambas as componentes não 
estão definidas), ser um campo gradiente é equivalente a ser zero a seguinte expressão:

> diff(vf[2],x)-diff(vf[1],y);

Assim o campo vetorial NÃO é um campo gradiente.

----

Outro exemplo:

> vf2:=[y^2-x^2,2*x*y+y];

> fieldplot(vf2,x=-2..2,y=-2..2);



> diff(vf2[2],x)-diff(vf2[1],y);

Assim este campo é gradiente de uma função. Como a função tem diff(f,x)=vf2[1] e diff
(f,y)=vf2[2] , construimos uma função dada por:

> f0:=int(vf2[1],x);

> f:=f0+int(vf2[2]-diff(f0,y),y);

e note:

> grad(f,[x,y]); vf2;

(O comando do Maple grad exige que você de uma lista de variáveis em ordem, assim 
você tem alguma flebilidade em nomear variaveis).



INTEGRAIS DE LINHA

Discutimos dois tipos de integrais de linha aqui... em qualquer caso o passo mais 
importante é

parametrizar a curva sobre a qual você deseja tomar a integral.

Naturalmente, se a curva eh o grafico de uma função y=f(x) , a parametrização mais 
simples é x=t e y=f(t) . Por exemplo, a parábola y=1-x^2 entre x=-1 e x=1 pode ser 
parametrizada:

> parab:=[t,1-t^2,t=-1..1];

A parte da variação do parametro é muito importante.

Podemos plotar a parabola:

> plot(parab);

Vamos integrar a função

> g:=sin(x)*y^2;



sobre a parabola. Para fazer isto, precisamos substituir a parametrização da curva na 
função:

> pg:=subs(x=parab[1],y=parab[2],g);

e integrar contra ds (que é sqrt(dx^2+dy^2)) :

> Int(pg*sqrt(diff(parab[1],t)^2+diff(parab[2],t)^2),parab[3]);

Esta integral não pode ser dada pelo Maple numa forma fechada, mas podemos avaliar:

> evalf(");

Um procedimento para fazer integrais de linha deste tipo (que assume as variaveis x,y e t 
como foi feito acima), é:

> Lineint1:=proc(fn,curve)

> Int(subs(x=curve[1],y=curve[2],fn)*sqrt(diff(curve[1],t)^2+ diff(curve[2],t)^2),curve[3]);

> end;

Vamos testar:

> Lineint1(g,parab);

Entao para avaliar a integral ou usamos "value" ou "evalf".

> evalf(Lineint1(g,parab));



O segundo e mais comum tipo de integral de linha é a integral de um campo vetorial 
contra dr , onde r é a função vetorial de t que descreve a curva ...

Um exemplo tipico -- integrar vf dot dr ao longo de nossa parabola, onde vf eh o campo 
de vetores do primeiro exemplo:

> vf;

Observe, podemos obter um desenho da curva e do campo vetorial juntos, assim podemos 
obter uma ideia do que a resposta pode ser (pelo menos o sinal):

> with(plots,display);

> F:=fieldplot(vf,x=-2..2,y=-2..2):

> G:=plot(parab):

> display({F,G});

Como o campo de vetores é sempre contra a direção da curva, esperamos que a integral de 
linha será negativa:



Para avaliar a integral de linha, nós precisamos substituir a parametrização no campo de 
vetores, e então integrar o produto interno:

> vpar:=subs(x=parab[1],y=parab[2],vf);

> Int(dotprod(vpar,diff([parab[1],parab[2]],t)),parab[3]);

> value(");

Um procedimento em Maple para fazer istos (assumindo x,y and t com seus papeis 
usuais):

> Lineint2:=proc(vf,curve)

> Int(dotprod(subs(x=curve[1],y=curve[2],vf),diff([curve[1], curve[2]],t)),curve[3]);

> end;

Testando:

> Lineint2(vf,parab);

Vamos integrar vf sobre o circulo de raio 2 centrado na origem, onde o sentido é o 
contrario ao relogio (primeiro vamos olhar):

> circ:=[2*cos(t),2*sin(t),t=0..2*Pi];

> G:=plot(circ):



> display({F,G});

Desta vez o campo de vetores está a direção ao longo da curva, assim esperamos que o 
resultado seja positivo:

> Lineint2(vf,circ);

> value(");

O que você espera para vf2 em torno do circulo? 

> vf2;

> F:=fieldplot(vf2,x=-2..2,y=-2..2):

> display({F,G});



> Lineint2(vf2,circ);

> value(");

> 


